Interfacing the TC77 Thermal Sensor to a PICmicro® Microcontroller

Author: Steve Bible and Jim Lepkowski
Microchip Technology Inc.

INTRODUCTION

Silicon temperature sensors offer an easy-to-use alternative to traditional temperature sensors, such as thermocouple, thermistors and RTDs. The TC77 SPI™ Thermal Silicon Sensor is especially suited for embedded systems, due to its SPI interface. This serves to provide a straight-forward and easy way to interface to a microcontroller. This application note will discuss system integration, firmware implementation and PCB layout techniques for the TC77 in an embedded system.

Microchip has developed a hardware platform called the PICkit™ 1 FLASH Starter Kit, allowing the designer to quickly begin their system development. Additionally, Microchip has developed a TC77 PICtail™ Daughter Board that interfaces directly to the PICkit 1 FLASH Starter Kit. These two boards are used to demonstrate the techniques for integrating the TC77 into an embedded systems environment. Both of these development boards are available on the Microchip web site at www.microchip.com.

The TC77 SPI™ Thermal Sensor PICtail™ daughter board is designed to demonstrate the ease of integrating a digital silicon IC temperature sensor to a PICmicro® microcontroller unit (MCU). The TC77 PICtail daughter board plugs into the PICkit 1 FLASH Starter Kit expansion header J3, as shown in Figure 1. The PICkit 1 FLASH Starter Kit is a low-cost development kit with an easy-to-use interface for programming Microchip's 8-pin and 14-pin FLASH family of microcontrollers.

The TC77 demonstration is designed to measure and display temperature in binary coded decimal (BCD) with the PICkit 1 LEDs. The TC77 is a CMOS silicon digital temperature sensor particularly suited for low cost and small form-factor applications. Temperature data is converted from the internal thermal sensing element and made available as a 13-bit two's compliment digital word. The TC77 offers many system-level advantages, including the integration of the temperature sensor and signal conditioning circuitry on a single chip that is connected to the PICkit 1 through the SPI compatible interface.

Gerber files for the Printed Circuit Board (PCB), source code and hex file to program a PIC16F676 are included in the companion zip file "00913.zip".



FIGURE 1: Block Diagram of the TC77 Thermal Sensor Demonstration.

TC77 FUNCTIONAL DESCRIPTION

The TC77 consists of an internal diode temperature sensor, a 13-bit Delta-Sigma, Analog-to-Digital Converter (ADC), three digital registers and a SPI compatible interface. The SPI compatible interface provides for serial communication with microcontrollers, such as a PICmicro microcontroller. Figure 2 provides a simplified block diagram of the TC77 sensor.

The temperature measurement data is stored in the Temperature register, while the Configuration register is used to select the operating mode of the sensor. The Manufacturer's Identification (ID) register is used to identify the sensor as a Microchip component. Table 1 provides the bit definitions of the TC77 registers.

Operating Modes

The user-configured operating modes of the TC77 include a Continuous Temperature and a Shutdown mode that are selected via the Configuration register. In the Continuous Temperature mode, an ADC conversion is performed approximately every 300 ms, with the data being stored in the Temperature register. If a Temperature register read operation is requested while an ADC conversion is in progress, the previously completed ADC conversion data will be outputted via the sensor's serial I/O port.

Shutdown mode can be used to minimize the power consumption of the TC77 sensor when active temperature monitoring is not required. While Shutdown mode disables the temperature conversion circuitry, the SPI compatible interface remains active. The current consumption of the sensor will be less than 1 µA when Shutdown mode is activated.



FIGURE 2: *Block Diagram of the TC77 Thermal Sensor.*

TABLE 1: TC77 DIGITAL REGISTERS

Register	Bit 15	Bit 14	Bit 13	Bit 12	Bit 11	Bit 10	Bit 9	Bit 8	Bit 7	Bit 6	Bit 5	Bit 4	Bit 3	Bit 2	Bit 1	Bit 0	Value at Power-up/ Reset
Configuration (Read/Write)	C15	C14	C13	C12	C11	C10	C9	C8	C7	C6	C5	C4	C3	C2	C1	C0	Continuous Temperature Conversion Mode **
Temperature (Read Only)	T15 (2^8) ***	T14 (2^7)	T13 (2^6)	T12 (2^5)	T11 (2^4)	T10 (2^3)	T9 (2^3)	T8 (2^1)	T7 (2^0)	T6 (2^{-1})	T5 (2^{-2})	T4 (2^{-3})	T3 (2^{-4})	T2 *	T1 x	T0 x	Temp. = -2°C
Manufacturer ID (Read Only)	0	1	0	1	0	1	0	0	0	0	0	0	0	0	x	x	Bit 15 to Bit 8 = 54 hex

Legend: * Temperature Bit 2 = `0` during power-up; otherwise, bit 2 = `1`
** C15:C0 = `XXXX/XXXX 1111/1111` (Shutdown mode)
C15:C0 = `XXXX/XXXX 0000/0000` (Continuous Conversion mode)
*** Temperature Register Bit 15 is the sign bit. If Bit 15 is equal to '`1`', the temperature is negative (T < 0°C). If Bit 15 is equal to '`0`', the temperature is positive (T ≥ 0°C).

SPI Compatible Interface

The TC77's SPI compatible interface consists of the Chip Select ($\overline{CS}$), Serial Clock (SCK) and bidirectional Serial Input/Output (SI/O) data signals. Figure 3 provides a timing diagram of a read operation of the Temperature register.

Communication with the TC77 is initiated when the $\overline{CS}$ goes to a logic '0'. The SI/O signal then transmits the first bit of data. The SCK input is provided by the PICmicro microcontroller and data is transferred on the rising edge of SCK. The SI/O line is then tri-stated once 14 bits of data have been transmitted.

The $\overline{CS}$ input is used to select the TC77 when multiple devices are connected to the SPI lines. The $\overline{CS}$ line is also used to synchronize the data, which is written to, or read from, the device when $\overline{CS}$ is equal to a logic '0'. The SCK input is disabled when $\overline{CS}$ is a logic '1'. The falling edge of the $\overline{CS}$ line initiates communication, while the rising edge of $\overline{CS}$ completes the communication.

Figure 4 provides a timing diagram of a multi-byte communication operation consisting of a read of the Temperature Data register, followed by a write to the Configuration register. The first 16 SCK pulses are used to transmit the TC77's temperature data to the microcontroller. The second group of 16 SCK pulses are used to receive the microcontroller command to place the TC77 either in Shutdown or Continuous Temperature Conversion mode. Note that the TC77 is in the Continuous Temperature Conversion mode at power-up.

The data written to the TC77's Configuration register should be either all 0's or all 1's, corresponding to either the Continuous Temperature Conversion or Shutdown mode, respectively. The TC77 is in Shutdown mode when bits C0 to C7 are all equal to 1's. The TC77 will be in the Continuous Conversion mode if a '0' in any bit location from C0 to C7 is written to the Configuration register.

Temperature Data Format

The TC77's temperature data is represented by a 13-bit two's complement digital word as shown in Table 1 and Table 2. The Least Significant bit (LSb) is equal to 0.0625°C. Note that the last two bits (bit T0 and T1) are tri-stated and are represented as a logic '1' in the table. bit T2 is set to logic '1' after the completion of the first temperature conversion following a power-up or voltage reset event.

Listed below is an example of the TC77's Temperature Register bit definition for a temperature of 85.125°C.

Example:

Temperature	=	85.125°C
Temperature Register	=	00101010 10010111b
	=	$2^6 + 2^4 + 2^2 + 2^0 + 2^{-3}$
	=	64 + 16 + 4 + 1 + 0.125
	=	85.125°C

TABLE 2: TC77 TEMPERATURE OUTPUT DATA

Temperature	Binary Bit 15 … Bit 0
+125°C	0011 1110 1000 0111
+25°C	0000 1100 1000 0111
+0.0625°C	0000 0000 0000 1111
0°C	0000 0000 0000 0111
-0.0625°C	1111 1111 1111 1111
-25°C	1111 0011 1000 0111
-55°C	1110 0100 1000 0111



FIGURE 3: *Temperature Read Timing Diagram.*



TEMPERATURE REGISTER

Notes:

1. Bit T2 = `0` during power-up for the first temperature conversion.
2. Bit T2 = `1` after the completion of the first temperature conversion following power-up or a reset event.

CONFIGURATION REGISTER

Notes:

1. XX00 = Continuous Conversion Mode
2. XXFF = Shutdown Mode

FIGURE 4: *Temperature Read Followed By A Write To The Configuration Register Timing Diagram.*

TC77 Application Guidelines

It is recommended that a decoupling capacitor of 0.1 µF to 1 µF be provided between the power supply and ground pins to provide effective noise protection to the sensor. Also, the user should select a TC77 sensor that has a calibration voltage that is as close as possible to the system voltage on the PCB. The TC77's temperature accuracy is tested and calibrated at either 3.3V or 5.0V, with the accuracy being degraded if a different voltage is used than the calibration V_{DD}. Please refer to the TC77 data sheet (DS20092) for further details on the specifications of the sensor.

Silicon digital temperature sensors measure temperature by monitoring the voltage of a diode located on the die. The TC77's substrate of the die is grounded and connected to the PCB's ground plane via a bonding wire and package lead. The ground pin provides a low-impedance thermal path between the die and the PCB, allowing the sensor to effectively monitor the temperature of the PCB board.

The thermal path between the top of the package to the ambient air, and between the bottom of the package and the PCB, is not as efficient because the plastic package functions as a thermal insulator. Thus the ambient air temperature (assuming that a large temperature gradient exists between the air and PCB) has only a small effect on the temperature measured by the temperature sensor.

TC77 PICTAIL DAUGHTER BOARD

The TC77 PICtail daughter board is plugged to the PICkit 1 FLASH Starter Kit via expansion header J3. Figure 5 shows a picture of the TC77 PICtail daughter board plugged into the PICkit 1 FLASH Starter Kit. For more information on the PICkit 1 FLASH Starter Kit, refer to the "PICkit 1 FLASH Starter Kit User's Guide" (DS40051).

The TC77 PICtail daughter board consists of a TC77 temperature sensor and a bypass capacitor. The bypass capacitor (C_1) is used to provide noise immunity on the +5 VDC power supply. Figure 6 shows a schematic of the board, while Figure 7 provides a layout drawing of the PCB. The Bill of Materials (BOM) is given in Table 3. Gerber files for the TC77 PICtail daughter board are available in the companion zip file "00913.zip".

TABLE 3: TC77 THERMAL SENSOR PICtail™ DAUGHTER BOARD (112-00004) BILL OF MATERIALS (BOM)

Component	Symbol	Part Number
Temperature Sensor 5V, SOT-23	U1	TC77-5.0MCT
Capacitor, 0.1 µF, ±10%, 25V	C_1	ECJ-2VBIE104K (Panasonic®)
Connector, 14-pin	P1	800-99-014-20-001 (Mill-Max™)
Connector, 14-pin	P2, P3	Not Used
Printed Circuit Board	—	102-00004



FIGURE 5: *TC77 PICtail™ Daughter Board and PICkit™ 1 FLASH Starter Kit.*



FIGURE 6: *TC77 PICtail™ Daughter Board Schematic.*



FIGURE 7: *TC77 PICtail™ Daughter Board PCB Layout.*

TC77 Interface Software

A flow diagram for the PICkit 1 software is given in Figure 8. The TC77 thermal sensor is read by the PICmicro microcontroller. Bit T2 of the Temperature register is tested to ensure that the TC77 sensor is powered up and ready. The value read from the TC77 is right-adjusted in the register as a 12-bit temperature value in degrees Celsius. The temperature value is tested for a negative temperature reading by checking the status of bit T15. If the value is negative ($T < 0°C$), the state is saved in a flag bit and the value is 2's complemented.

The TC77's Temperature register provides a temperature measurement in Celsius. A provision in the software is provided to display the temperature in either Fahrenheit or Celsius by testing the status of the PICkit 1 push button switch (SW1). If SW1 is not depressed, the temperature value is converted to Fahrenheit. Otherwise, if the push button is depressed, the conversion routine is skipped and the data is displayed in Celsius. Finally, the temperature value is loaded into the LEDREG variable to be displayed on the LEDs by the DISPLAY subroutine.

The temperature measurement is displayed using the red LEDs designated as D0 through D7 located on the PICkit board. The ten's digit of the temperature data is represented by bits D7 to D4, with D7 being defined as the Most Significant bit (MSb). The one's digit is defined by bits D3 to D0, with D3 serving as the MSb. Table 4 provides a list of the LED lamp annunciation that corresponds to the BCD coding representation of the temperature measurement. For example, a temperature reading of 70°F will be displayed by illuminating LEDs D4, D5 and D6. If the SW1 push button is activated, a measurement of 21°C will be shown by illuminating LEDs D0 and D5. A fractional temperature is rounded up if the tenths digit is determined to be either 0.5°F or 0.5°C.

Fully documented source code and a hex file ready to program into a PIC16F676 is available in the companion zip file “00913.zip”.

TABLE 4: LED LAMP ANNUNCIATION

Binary	BCD	LED Annuciation			
		D7 D3	D6 D2	D5 D1	D4 D0
0000	0	OFF	OFF	OFF	OFF
0001	1	OFF	OFF	OFF	ON
0010	2	OFF	OFF	ON	OFF
0011	3	OFF	OFF	ON	ON
0100	4	OFF	ON	OFF	OFF
0101	5	OFF	ON	OFF	ON
0110	6	OFF	ON	ON	OFF
0111	7	OFF	ON	ON	ON
1000	8	ON	OFF	OFF	OFF
1001	9	ON	OFF	OFF	ON



FIGURE 8: *TC77 PICtail™ Program Flow Diagram.*



FIGURE 8: *TC77 PICtail™ Program Flow Diagram (Cont.).*

CONCLUSION

The TC77 SPI™ Thermal Sensor PICtail™ daughter board demonstrates the ease of integrating a digital silicon IC temperature sensor to a PICmicro microcontroller unit (MCU). The TC77 is a CMOS silicon digital temperature sensor that provides an accurate digital temperature measurement to solve thermal management problems. The TC77 sensors offer many system level advantages, including the integration of the sensor and the signal conditioning circuitry in a small IC package. This provides for easy system integration and minimizes the required PCB space, component count, and design time.

BIBLIOGRAPHY

1. AN871, "*Solving Thermal Measurement Problems Using the TC72 and TC77 Digital Silicon Temperature Sensors*", Jim Lepkowski, Microchip Technology Inc., DS00871, 2003.
2. "*PICkit™ 1 FLASH Starter Kit User's Guide*", Microchip Technology Inc., DS40051, 2003.
3. TC77 Data Sheet, "*Thermal Sensor with SPI™ Interface*", Microchip Technology Inc., DS20092, 2002.

APPENDIX A: SOURCE CODE

```
;   Filename:           TC77 PICtail.asm
;   Date:               December 17, 2003
;   File Version:       0.2
;   Assembled using:
;
;   Author:             Steven Bible
;   Company:            Microchip Technology Inc.
;
;
;---------------------------------------------------------------------
;
;   Files required:
;                       p16f676.inc
;
;
;---------------------------------------------------------------------
;
;   Program Description
;
;   This program demonstrates the Microchip TC77 Thermal Sensor with
;   SPI(tm) compatible interface using the PICkit(tm) 1 FLASH Starter Kit.
;   The temperature is read from the TC77 and displayed on LEDs
;   D0 through D7 in Binary Coded Decimal (BCD).
;
;---------------------------------------------------------------------

    list        p=16f676        ; list directive to define processor
    #include    <p16f676.inc>   ; processor-specific variable definitions

    errorlevel  -302            ; suppress message 302 from list file

; --------------------------------------------------------------------
; Configuration Bits (Section 9.1 Configuration Bits)
; --------------------------------------------------------------------
;
; Data Memory Code Protection bit:
; _CPD = Enabled
; _CPD_OFF = Disabled
;
; Program Memory Code protection:
; _CP = Enabled
; _CP_OFF = : Disabled
;
; Brown-out Detection Enable bit:
; _BODEN = Enabled
; _BODEN_OFF = Disabled
;
; GP3/MCLR pin function select:
; _MCLRE_ON = GP3/MCLR pin function is /MCLR
```

```
; _MCLRE_OFF = GP3/MCLR pin function is digital I/O,
;              /MCLR internally tied to Vdd
;
; Power-up Timer Enable bit:
; _PWRTE_ON = Enabled
; _PWRTE_OFF = Disabled
;
; Watchdog Timer Enable bit:
; _WDT_ON = Enabled
; _WDT_OFF = Disabled
;
; Oscillator Selction bits:
; _EXTRC_OSC_NOCLKOUT = CLKOUT function on GP4 pin, RC on GP5 pin.
; _EXTRC_OSC_CLKOUT = I/O function on GP4 pin, RC on GP5 pin.
; _INTRC_OSC_CLKOUT =  Internal oscillator, CLKOUT function on GP4 pin,
;                      I/O function on GP5 pin.
; _INTRC_OSC_NOCLKOUT = Internal oscillator, I/O function on GP4 and GP5 pins.
; _EC_OSC = I/O function on GP4 pin, CLKIN on GP5 pin.
; _HS_OSC = High speed crystal/resonator on GP4 and GP5 pins.
; _XT_OSC = Crystal/resonator on GP4 and GP5 pins.
; _LP_OSC = Low power crystal on GP4 and GP5 pins.
;
;
; ---------------------------------------------------------------------

    __CONFIG    _CPD_OFF & _CP_OFF & _BODEN & _MCLRE_OFF & _PWRTE_ON & _WDT_OFF &
_INTRC_OSC_NOCLKOUT

;----------------------------------------------------------------------
; Variables  (Section 2.2 Data Memory Organization)
;----------------------------------------------------------------------

    ; Data Memory Organization (Section 2.2)
    ;
    ; The data memory is partitioned into two banks which contain
    ; the General Purpose registers and the Special Function registers.
    ; The Special Function registers are located in the first 32
    ; locations of each bank. Register locations 0x20 to 0x5F (64 bytes)
    ; are General Purpose registers, implemented as static RAM and are
    ; mapped across both banks.
    ;
    ;   RP0 (STATUS<5>)
    ;    0  ->  Bank 0
    ;    1  ->  Bank 1
    ;
    ; Refer to Section 2.2 of the data sheet for the organization of
    ; the General Purpose Registers.

    ; Bank 0 General Purpose Registers

    cblock  0x20    ; File Address 0x20-0x5F (64 bytes)

        W_TEMP                  ; used for context saving
        STATUS_TEMP             ; used for context saving
        PCLATH_TEMP             ; used for context saving
        FSR_TEMP                ; used for context saving

        TEMP                    ; General Purpose Temporary register

        FLAG                    ; A byte of binary flags (see Defines below)

        TICK                    ; Tick counter

        ; LED Display on PICkit 1 Flash Starter Kit
```

```
        LEDREG                      ; LED Array Register
        LEDSTATE                    ; LED Array State Counter
        LEDDISP                     ; LED Array Display bit (which LED is lit)

        ; TC77 Thermal Sensor variables

        BIT_CNTR                    ; Bit counter

        TC77_HI                     ; TC77 Temperature register high byte
        TC77_LO                     ; TC77 Temperature register low byte

        TEMP_HI                     ; Temporary Register high byte
        TEMP_LO                     ; Temporary Register low byte

        ; Binary Coded Decimal (BCD) variables

        BCD_H                       ; BCD Hundreds
        BCD_T                       ; BCD Tens
        BCD_O                       ; BCD Ones

    endc


;----------------------------------------------------------------------
; Defines
;----------------------------------------------------------------------

    ;--------------------
    ; PORTA (Section 3.1)
    ;--------------------
    ; PORTA is an 6-bit wide, bidirectional port. The corresponding data
    ; direction register is TRISA. Setting a TRISA bit (= 1) will make
    ; the corresponding PORTA pin an input. Clearing a TRISA bit (= 0)
    ; will make the corresponding PORTA pin an output. The exception is
    ; RA3, which is input-only and whose TRIS bit will always read as a '1'.
    ;
    ; Function of PORTA pins depend on:
    ;   Configuration Bits (CONFIG) (Section 9.1)
    ;   Weak Pull-up Register (WPU) (Section 3.2.1)
    ;   Interrupt-on-change Register (IOCB) (Section 3.2.2)
    ;   Option Register (OPTION_REG) (Register 4-1)
    ;   TIMER1 Control Register (T1CON) (Register 5-1)
    ;   Comparator Control Register (CMCON) (Section 6.0)
    ;   A/D Control Register (ADCON0) (Section 7.0) (PIC16F676 Only)

#define POT     PORTA, 0            ; (Analog Input) Potentiometer RP1
#define RA1     PORTA, 1            ; (Digital Input/Output) LEDs D6, D7
#define RA2     PORTA, 2            ; (Digital Input/Output) LEDs D2, D3, D4, D5, D6, D7
#define SW1     PORTA, 3            ; (Digital Input Only) Push Button SW1
#define RA4     PORTA, 4            ; (Digital Input/Output) LEDs D0, D1, D2, D3
#define RA5     PORTA, 5            ; (Digital Input/Output) LEDs D0, D1, D4, D5

    ; Define for TRISA Register (Section 3.1)

    ;   PORTA Pins = xx543210
#define PORTATRIS  b'00111111'

    ;--------------------
    ; PORTC (Section 3.3)
    ; --------------------
    ; PORTC is a general purpose I/O port consisting of 6 bidirectional
    ; pins. The pins can be configured for either digital I/O or for analog
    ; input to an A/D converter. For specific information about individual functions
    ; such as the comparator or the A/D, refer to the appropriate section in the
    ; data sheet.
```

```
#define SCK     PORTC, 0            ; (Digital Output) Serial Clock
#define SIO     PORTC, 1            ; (Digital Input/Output) Serial I/O (initially set as input)
#define RC2     PORTC, 2            ; (Digital Input)
#define TC77_CS PORTC, 3            ; (Digital Output) TC77 Chip Select (active low)
#define RC4     PORTC, 4            ; (Digital Input)
#define RC5     PORTC, 5            ; (Digital Input)

    ; Define for TRISC Register (Section 3.3)

    ;   PORTC Pins = xx543210
#define PORTCTRIS  b'00110110'

    ;--------------------
    ; Program Defines
    ;--------------------

    ; Flags

#define TRIP     0              ; Tick counter trip flag
#define SIGN_BIT 1              ; temperature sign bit
#define C_F_DISP 2              ; Display in C or F

    ; LEDs

    ;    PORTA Pins = xx543210
#define LED0TRIS    b'00001111'
#define LED1TRIS    b'00001111'
#define LED2TRIS    b'00101011'
#define LED3TRIS    b'00101011'
#define LED4TRIS    b'00011011'
#define LED5TRIS    b'00011011'
#define LED6TRIS    b'00111001'
#define LED7TRIS    b'00111001'
#define LEDOFFTRIS  b'00111111'

    ;    PORTA Pins = xx543210
#define LED0ON      b'00010000'
#define LED1ON      b'00100000'
#define LED2ON      b'00010000'
#define LED3ON      b'00000100'
#define LED4ON      b'00100000'
#define LED5ON      b'00000100'
#define LED6ON      b'00000100'
#define LED7ON      b'00000010'


;----------------------------------------------------------------------
; Program Memory
;----------------------------------------------------------------------

    ; Program Memory Organization (Section 2.1)

        ORG     0x0000              ; RESET Vector

        nop                         ; for ICD use
        goto    MAIN                ; goto MAIN Program


        ORG     0x0004              ; Interrupt Vector

        movwf   W_TEMP              ; save W register
        swapf   STATUS, W           ; swap status to be saved into W
        bcf     STATUS, RP0         ; ---- Select Bank 0 -----
        movwf   STATUS_TEMP         ; save STATUS register
```

```
        movfw   PCLATH
        movwf   PCLATH_TEMP         ; save PCLATH_TEMP register
        movfw   FSR
        movwf   FSR_TEMP            ; save FSR_TEMP register

;-----------------------------------------
; Interrupt Service Routine (ISR) (Section 9.4)
;
; Description:
;
;-----------------------------------------

        bcf     INTCON, T0IF        ; clear TMR0 Interrupt Flag

        call    DISPLAY             ; Update LED Array (light LEDs)

        decf    TICK, F             ; decrement tick counter
        btfsc   STATUS, Z
         bsf    FLAG, TRIP

;-----------------------------------------

        movfw   PCLATH_TEMP         ; restore PCLATH_TEMP register
        movwf   PCLATH
        movfw   FSR_TEMP            ; restore FSR_TEMP register
        movwf   FSR
        swapf   STATUS_TEMP, W      ; swap status_temp into W, sets bank to original state
        movwf   STATUS              ; restore STATUS register
        swapf   W_TEMP, F
        swapf   W_TEMP, W           ; restore W register

        retfie


;-----------------------------------------------------------------------
; Initialize PICmicro(r) MCU (PIC16F630/676)
;-----------------------------------------------------------------------

INITIALIZE

; Disable global interrupts during initialization

        bcf     INTCON, GIE         ; disable global interrupts


;-----------------------------------------
; Calibrating the Internal Oscillator (Section 9.2.5.1)
; Oscillator Calibration Register (OSCCAL) (Section 2.2.2.7)
;
; A calibration instruction is programmed into the last location of
; program memory. This instruction is a RETLW XX, where the literal is
; the calibration value. The literal is placed in the OSCCAL register
; to set the calibration of the internal oscillator.

        bsf     STATUS, RP0         ; ---- Select Bank 1 -----

        call    0x3FF               ; retrieve factory calibration value
        movwf   OSCCAL              ; update register with factory cal value

        bcf     STATUS, RP0         ;---- Select Bank 0 -----


;-----------------------------------------
; PORTS A AND C (Section 3.0)
;
```

```
; Store PORTATRIS and PORTCTRIS values defined above into the
; TRISA and TRISC direction registers

        bsf     STATUS, RP0         ; ---- Select Bank 1 -----

        movlw   PORTATRIS
        movwf   TRISA               ; Write to TRISA register

        movlw   PORTCTRIS
        movwf   TRISC               ; Write to TRISC register

        bcf     STATUS, RP0         ;---- Select Bank 0 -----


;----------------------------------------
; Comparator Module (Section 6.0)
;
; The PIC16F630/676 devices have one analog comparator. The inputs to
; the comparator are multiplexed with the RA0 and RA1 pins. There is
; an on-chip Comparator Voltage Reference that can also be applied to
; an input of the comparator. In addition, RA2 can be configured as
; the comparator output. The Comparator Control register (CMCON)
; contains bits to control the comparator. The Voltage Reference
; Control register (VRCON) controls the voltage reference module.

        ; Comparator Configuration (Figure 6-2)
;        bcf     CMCON, CINV         ; Comparator Output Inversion: not inverted
;        bcf     CMCON, COUT         ; Comparator Output bit: Vin+ < Vin-
;        bcf     CMCON, CIS          ; Comparator Input Switch: Vin- connects to Cin-

        ; CM2:CM0 = 111 - Comparator Off (lowest power)
        bsf     CMCON, CM2          ; Comparator Mode bit 2
        bsf     CMCON, CM1          ; Comparator Mode bit 1
        bsf     CMCON, CM0          ; Comparator Mode bit 0

        ; VRCON (Register 6-2)
        bsf     STATUS, RP0         ; ---- Select Bank 1 -----

        bcf     VRCON, VREN         ; CVref circuit: powered down, no Idd drain

;        bcf     VRCON, VRR          ; CVref Range Selection: High Range

;        bcf     VRCON, VR3          ; CVref value selection bit 3
;        bcf     VRCON, VR2          ; CVref value selection bit 2
;        bcf     VRCON, VR1          ; CVref value selection bit 1
;        bcf     VRCON, VR0          ; CVref value selection bit 0

        bcf     STATUS, RP0         ;---- Select Bank 0 -----


;----------------------------------------
; Analog-to-Digital Converter (A/D) Module (Section 7.0) (PIC16F676 Only)
;
; The analog-to-digital converter (A/D) allows conversion of an analog
; input signal to a 10-bit binary representation of that signal. The
; PIC16F676 has eight analog inputs multiplexed into one sample and hold
; circuit. There are two registers to control the functions of the A/D
; module:
;   A/D Control Register 0 (ADCON0)
;   A/D Control Register 1 (ADCON1)
;   Analog Select Register (ANSEL)
;
; Note: When using PORTA or PORTC pins as analog inputs, ensure the
;       TRISA or TRISC register bits are set (= 1) for input.
```



```
        bcf     ADCON0, ADFM        ; A/D Result Formed: left justified
        bcf     ADCON0, VCFG        ; Voltage Reference: Vdd

        bsf     STATUS, RP0         ; ---- Select Bank 1 -----

        ; select A/D Conversion Clock Source: Fosc/8
        bcf     ADCON1, ADCS2       ; A/D Conversion Clock Select bit 2
        bcf     ADCON1, ADCS1       ; A/D Conversion Clock Select bit 1
        bsf     ADCON1, ADCS0       ; A/D Conversion Clock Select bit 0

        ; select GPIO pins that will be analog inputs: RA0/AN0
        bcf     ANSEL, ANS7         ; Analog Select RC3/AN7: digital I/O
        bcf     ANSEL, ANS6         ; Analog Select RC2/AN6: digital I/O
        bcf     ANSEL, ANS5         ; Analog Select RC1/AN5: digital I/O
        bcf     ANSEL, ANS4         ; Analog Select RC0/AN4: digital I/O
        bcf     ANSEL, ANS3         ; Analog Select RA3/AN3: digital I/O
        bcf     ANSEL, ANS2         ; Analog Select RA2/AN2: digital I/O
        bcf     ANSEL, ANS1         ; Analog Select RA1/AN1/Vref: digital I/O
        bsf     ANSEL, ANS0         ; Analog Select RA0/AN0: analog input

        bcf     STATUS, RP0         ;---- Select Bank 0 -----

        bcf     ADCON0, ADON        ; ADC is shut-off and consumes no operating current


;----------------------------------------
; TIMER1 Module with Gate Control (Section 5.0)
;
; The TIMER1 Control Register (T1CON) is used to enable/disable TIMER1
; and select various features of the TIMER1 module.

        bcf     T1CON, TMR1ON       ; TIMER1: stopped

        bcf     T1CON, TMR1CS       ; TIMER1 Clock Source Select: Internal Clock (Fosc/4)

       bcf     T1CON, NOT_T1SYNC   ; TIMER1 External Clock Input Sync Control: Syncronize external
clock input

        ; T1OSCEN only if INTOSC without CLKOUT oscillator is active, else ignored
        bcf     T1CON, T1OSCEN      ; LP Oscillator Enable Control: LP oscillator off

        ; TIMER1 Input Prescale Select: 1:1
        bcf     T1CON, T1CKPS1      ; TIMER1 Input Clock Prescale Select bit 1
        bcf     T1CON, T1CKPS0      ; TIMER1 Input Clock Prescale Select bit 0

        ; TMR1GE only if TMR1ON = 1, else ignored
        bcf     T1CON, TMR1GE       ; TIMER1 Gate Enable: on


;----------------------------------------
; PORTA Weak Pull-up Register (WPUA) (Section 3.2.1)
;
; Each of the PORTA pins, except RA3, has an individually configurable
; weak internal pull-up. Control bits WPUAx enable or disable each
; pull-up. Refer to Register 3-1. Each weak pull-up is automatically
; turned off when the port pin is configured as an output. The pull-ups
; are disabled on a Power-on Reset by the /RAPU bit (see OPTION Register
; below).

        bsf     STATUS, RP0         ; ---- Select Bank 1 -----

;    PORTA Pins = xx54x210
        movlw   B'00000000'         ; no pull-ups enabled
        movwf   WPUA
```

```
        bcf     STATUS, RP0         ;---- Select Bank 0 -----
;----------------------------------------
; OPTION Register (OPTION_REG) (Section 2.2.2.2)
; TIMER0 Module (Section 4.0)
;
; The OPTION_REG contains control bits to configure:
;   Weak pull-ups on GPIO (see also WPU Register above)
;   External RA2/INT interrupt
;   TMR0
;   TMR0/WDT prescaler

        bsf     STATUS, RP0         ; ---- Select Bank 1 -----

        bsf     OPTION_REG, NOT_GPPU ; PORTA pull-ups: disabled

        bsf     OPTION_REG, INTEDG  ; Interrupt Edge: on rising edge of RA2/INT pin

        bcf     OPTION_REG, T0CS    ; TMR0 Clock Source: internal instruction cycle (CLKOUT)
        bcf     OPTION_REG, T0SE    ; TMR0 Source Edge: increment low-to-high transition on GP2/
T0CKI pin

        bcf     OPTION_REG, PSA     ; Prescaler Assignment: assigned to TIMER0

        ; TMR0 Prescaler Rate: 1:8
        bcf     OPTION_REG, PS2     ; Prescaler Rate Select bit 2
        bsf     OPTION_REG, PS1     ; Prescaler Rate Select bit 1
        bcf     OPTION_REG, PS0     ; Prescaler Rate Select bit 0

        bcf     STATUS, RP0         ;---- Select Bank 0 -----


;----------------------------------------
; PORTA Interrupt-on-Change Register (IOCA) (Section 3.2.2)
;
; Each of the PORTA pins is individually configurable as an interrupt-
; on-change pin. Control bits IOCAx enable or disable the interrupt
; function for each pin. Refer to Register 3-4. The interrupt-on-change
; is disabled on a Power-on Reset.
;
; Note: Global interrupt enables (GIE and GPIE) must be enabled for
;       individual interrupts to be recognized.

        bsf     STATUS, RP0         ; ---- Select Bank 1 -----

;     GPIO Pins = xx54x210
        movlw   B'00000000'
        movwf   IOCA                ; Interrupt-on-change disabled

        bcf     STATUS, RP0         ;---- Select Bank 0 -----


;----------------------------------------
; Peripheral Interrupt Enable Register (PIE1) (Section 2.2.2.4)
;
; The PIE1 register contains peripheral interrupt enable bits.
;
; Note: The PEIE bit (INTCON<6>) must be set to enable any
;       peripheral interrupt.

        bsf     STATUS, RP0         ; ---- Select Bank 1 -----

        bcf     PIE1, EEIE          ; EE Write Complete Interrupt: disabled
        bcf     PIE1, ADIE          ; A/D Converter Interrupt (PIC12F675 Only): disabled
        bcf     PIE1, CMIE          ; Comparator Interrupt: disabled
        bcf     PIE1, TMR1IE        ; TMR1 Overflow Interrupt: disabled
```

```
        bcf     STATUS, RP0         ;---- Select Bank 0 -----


;---------------------------------------
; Interrupt Control Register (INTCON) (Section 2.2.2.3)
;
; The INTCON register contains enable and disable flag bits for TMR0
; register overflow, GPIO port change and external GP2/INT pin
; interrupts.

        bsf     INTCON, T0IE        ; TMR0 Overflow Interrupt: ENABLED
        bcf     INTCON, INTE        ; RA2/INT External Interrupt: disabled
        bcf     INTCON, RAIE        ; Port Change Interrupt: disabled

        bcf     INTCON, PEIE        ; Peripheral Interrupts: disabled
                                    ; (EEI, ADI, CMI, TMR1I)

        bcf     INTCON, GIE         ; Global Interrupts: disabled

        return                      ; return from INITIALIZE

; end INITIALIZE

;----------------------------------------------------------------------
; Subroutine: DATA_EEPROM_READ
;
; Description: To read an EEPROM data memory location, the address is
;   written to the EEADR register and set control bit RD (EECON1<0>) to
;   initiate a read. Data is available in the EEDATA register the next
;   clock cycle.
;
; Constants: none
;
; Global Variables: none
;
; Initialization: W contains EEPROM address (EEADR) to be read
;
; Output: W contains EEPROM data (EEDATA)
;
;----------------------------------------------------------------------

DATA_EEPROM_READ

        bsf     STATUS, RP0         ; ---- Select Bank 1 -----

        movwf   EEADR               ; move EEPROM address in W to EEADR
        bsf     EECON1, RD          ; initiate EEPROM read
        movf    EEDATA, W           ; move data to W

        bcf     STATUS, RP0         ; ---- Select Bank 0 -----

        return


;----------------------------------------------------------------------
; Subroutine: DATA_EEPROM_WRITE
;
; Description: To write an EEPROM data memory location, the address is
;   written to the EEADR register, data to the EEDATA register, then
;   execute a required sequence of instructions.
;
; CAUTION: Interrupts are disable and then re-enabled during this
;          subroutine
;
```



```
; Constants: none
;
; Global Variables: none
;
; Initialization: Address = EEADR, Data = EEDATA
;
; Output: none
;
;-----------------------------------------------------------------------

DATA_EEPROM_WRITE

        bsf     STATUS, RP0         ; ---- Select Bank 1 -----

        bsf     EECON1, WREN        ; EEPROM Write Enable: allow write cycles
;        bcf     INTCON, GIE         ; disable global interrupts
                                    ; *** required sequence, do not alter ***
        movlw   0x55
        movwf   EECON2
        movlw   0xAA
        movwf   EECON2
        bsf     EECON1, WR          ; initiate EEPROM write
                                    ; *** end required sequence ***

        btfsc   EECON1, WR          ; has write completed?
        goto    $-1

;        bsf     INTCON, GIE         ; enable global interrupts
        bcf     EECON1, WREN        ; EEPROM Write Enable: inhibit write cycles

        bcf     STATUS, RP0         ; ---- Select Bank 0 -----

        return


;-----------------------------------------------------------------------
; Subroutine: READ_ANALOG_AN0
;
; Description: Read analog channel 0 (AN0).
;
; Constants: none
;
; Global Variables: none
;
; Initialization: none
;
; Output: ADRESH and ADRESL contain 10-bit A/D result justified
;   according to ADCON0, ADFM bit.
;
;-----------------------------------------------------------------------

READ_ANALOG_AN0

        bsf     ADCON0, ADON        ; Turn on ADC module

        bcf     ADCON0, CHS1        ; select analog channel AN0
        bcf     ADCON0, CHS0

        ; After selecting a new channel, allow for sufficent sample time.
        ; The amount of sample time depends on the charging time of the
        ; internal charge-holding capacitor (Section 7.2).

        movlw   D'6'                ; At 4 MHz, a 22 us delay
        movwf   TEMP                ; (22us = 2us + 6 * 3us + 1us)
        decfsz  TEMP, F
```

```
        goto    $-1

        bsf     ADCON0, GO          ; start A/D conversion

        btfsc   ADCON0, GO          ; has A/D conversion completed?
        goto    $-1

        bcf     ADCON0, ADON        ; Turn off ADC module (consumes no operating current)

        return

;-----------------------------------------------------------------------
; Subroutine: DISPLAY
;
; Description: Displays Value Stored In LEDREG On LED Array
;   1 LED is displayed during each call
;   D7..D4 LED'S show most significant nibble
;   D3..D0  LED'S show least significant nibble
;
; Constants:
;
; Global Variables: LEDREG, LEDDISP, LEDSTATE
;
; Initialization:
;
; Output:
;
;-----------------------------------------------------------------------

DISPLAY
        clrf    PORTA               ; turn off all LED's

        bcf     STATUS, C           ; clear the carry bit
        rlf     LEDDISP, F          ; rotate left the LED displayed bit
        btfsc   STATUS, C           ; was the bit rotated into carry?
         rlf    LEDDISP, F          ;  yes, put it back into bit 0

        incf    LEDSTATE, F         ;  no, increment LED State

        movfw   LEDREG              ; get LED Register, should the LED be lit?
        andwf   LEDDISP, W
        btfsc   STATUS, Z
         return                     ; bit was a zero, do not light and return

        movfw   LEDSTATE            ; Mask bits (should be only 8 states)
        andlw   B'00000111'

        addwf   PCL, F

        goto    LITELED0
        goto    LITELED1
        goto    LITELED2
        goto    LITELED3
        goto    LITELED4
        goto    LITELED5
        goto    LITELED6
        goto    LITELED7

LITELED0
        bsf     STATUS, RP0         ; ---- Select Bank 1 -----
        movlw   LED0TRIS
        movwf   TRISA
        bcf     STATUS, RP0         ; ---- Select Bank 0 -----
        movlw   LED0ON
        movwf   PORTA
```



```
        return

LITELED1
        bsf     STATUS, RP0         ; ---- Select Bank 1 -----
        movlw   LED1TRIS
        movwf   TRISA
        bcf     STATUS, RP0         ; ---- Select Bank 0 -----
        movlw   LED1ON
        movwf   PORTA
        return

LITELED2
        bsf     STATUS, RP0         ; ---- Select Bank 1 -----
        movlw   LED2TRIS
        movwf   TRISA
        bcf     STATUS, RP0         ; ---- Select Bank 0 -----
        movlw   LED2ON
        movwf   PORTA
        return

LITELED3
        bsf     STATUS, RP0         ; ---- Select Bank 1 -----
        movlw   LED3TRIS
        movwf   TRISA
        bcf     STATUS, RP0         ; ---- Select Bank 0 -----
        movlw   LED3ON
        movwf   PORTA
        return

LITELED4
        bsf     STATUS, RP0         ; ---- Select Bank 1 -----
        movlw   LED4TRIS
        movwf   TRISA
        bcf     STATUS, RP0         ; ---- Select Bank 0 -----
        movlw   LED4ON
        movwf   PORTA
        return

LITELED5
        bsf     STATUS, RP0         ; ---- Select Bank 1 -----
        movlw   LED5TRIS
        movwf   TRISA
        bcf     STATUS, RP0         ; ---- Select Bank 0 -----
        movlw   LED5ON
        movwf   PORTA
        return

LITELED6
        bsf     STATUS, RP0         ; ---- Select Bank 1 -----
        movlw   LED6TRIS
        movwf   TRISA
        bcf     STATUS, RP0         ; ---- Select Bank 0 -----
        movlw   LED6ON
        movwf   PORTA
        return

LITELED7
        bsf     STATUS, RP0         ; ---- Select Bank 1 -----
        movlw   LED7TRIS
        movwf   TRISA
        bcf     STATUS, RP0         ; ---- Select Bank 0 -----
        movlw   LED7ON
        movwf   PORTA
        return
```

```
;---------------------------------------------------------------------
; Subroutine: READ_TC77_TEMP
;
; Description:
;   The TC77 Temperature register is a 16-bit read-only register.
;   The temperature data format is a 13-bit two's complement digital
;   word (bits 15:3). The Least Significant bit (LSb) is equal to
;   0.0625 degrees C. Bit 2 is set to a logic '1' after the completion
;   of the first temperature conversion following a power-up or reset
;   event. Bits 1:0 are tri-stated.
;
; Constants:
;   None
;
; Global Variables:
;   TC77_HI = TC77 Temperature Register High Byte
;   TC77_LO = TC77 Temperature Register Low Byte
;
; Initialization:
;   The TC77_SIO TRIS bit is assumed to be set for input (=1)
;
; Output:
;   TC77_HI and TC77_LO contain the 16-bit Temperature Register value
;
;---------------------------------------------------------------------

READ_TC77_TEMP

        movlw   D'16'               ; set bit counter to 16
        movwf   BIT_CNTR

        bcf     SCK                 ; set SCK low
        bcf     TC77_CS             ; enable TC77 --> chip select low

READ_TC77_TEMP_LOOP

        bsf     SCK                 ; SCK rising edge

        btfsc   SIO                 ; read bit, if bit is set
         bsf     STATUS, C          ;  set carry bit
        btfss   SIO                 ; if bit is clear
         bcf     STATUS, C          ;  clear carry bit

        bcf     SCK                 ; set SCK low

        rlf     TC77_LO, F          ; rotate carry bit left into TC77_LO and _HI
        rlf     TC77_HI, F

        decfsz  BIT_CNTR, F         ; is reading the Temperature Register complete?
         goto   READ_TC77_TEMP_LOOP

        bsf     TC77_CS             ; disable TC77 --> chip select high

        return


;---------------------------------------------------------------------
;---------------------------------------------------------------------
; Main Program
;---------------------------------------------------------------------
;---------------------------------------------------------------------
```

```
MAIN

;---------------------------------------
; Initialize PICmicro® MCU
;---------------------------------------

        call    INITIALIZE

;---------------------------------------
; Initialize Variables
;---------------------------------------

        bcf     FLAG, TRIP          ; clear tick counter trip flag.

        bsf     TC77_CS             ; disable TC77 --> chip select high

        clrf    LEDREG              ; initialize the LED display routine
        clrf    LEDSTATE
        movlw   D'1'
        movwf   LEDDISP

        bsf     INTCON, GIE         ; enable global interrupts

MAINLOOP

; tick counter expired?

        btfss   FLAG, TRIP
         goto   MAINLOOP            ; no, loop
        bcf     FLAG, TRIP          ; clear tick counter trip flag.

; read temperature from TC77

        bcf     INTCON, GIE         ; disable global interrupts
        call    READ_TC77_TEMP
        bsf     INTCON, GIE         ; enable global interrupts

; is temperature conversion complete?

        ; check bit 2
        ; if not complete, read TC77 again
        ; (be sure to include code in the event bit 2 is never true)

        btfss   TC77_LO, 2
         goto   MAINLOOP

; right adjust 13-bit 2's complement temperature value into TC77_HI:TC77_LO

        bcf     STATUS, C           ; clear carry bit
        rrf     TC77_HI, F          ; rotate right TEMP_HI:TEMP_LO 3 bits
        rrf     TC77_LO, F
        bcf     STATUS, C           ; clear carry bit
        rrf     TC77_HI, F
        rrf     TC77_LO, F
        bcf     STATUS, C           ; clear carry bit
        rrf     TC77_HI, F
        rrf     TC77_LO, F

; if temperature is negative, save the sign bit and complement

        btfsc   TC77_HI, 4
         bsf     FLAG, SIGN_BIT
        btfss   TC77_HI, 4
         bcf     FLAG, SIGN_BIT
```

```
        btfss   FLAG, SIGN_BIT
         goto   ML00                ; temperature is positive, jump ahead

        bsf     TC77_HI, 7          ; sign extend bits 15:13 in TEMP_HI
        bsf     TC77_HI, 6
        bsf     TC77_HI, 5

        comf    TC77_HI, F          ; 2's complement
        comf    TC77_LO, F
        incf    TC77_LO, F
        btfsc   STATUS, C
        incf    TC77_HI, F

; display temperature in F (no push button press) or C (push button pressed)
ML00
        btfss   SW1                 ; is push button SW1 pressed?
         goto   ML20                ;  no, jump ahead

; to convert C to F:
; multiply temperature by 9

        movfw   TC77_HI             ; move TC77_HI:TC77_LO to TEMP_HI:TEMP_LO
        movwf   TEMP_HI             ; (save original temperature in TC77_HI:TC77_LO)
        movfw   TC77_LO
        movwf   TEMP_LO

        ; left shift 3 (multiply by 8)

        bcf     STATUS, C           ; clear carry bit
        rlf     TEMP_LO, F          ; rotate left TEMP_HI:TEMP_LO 3 bits
        rlf     TEMP_HI, F
        bcf     STATUS, C           ; clear carry bit
        rlf     TEMP_LO, F
        rlf     TEMP_HI, F
        bcf     STATUS, C           ; clear carry bit
        rlf     TEMP_LO, F
        rlf     TEMP_HI, F

        ; add TC77_HI:TC77_LO (multiply by 9)

        movfw   TC77_LO
        addwf   TEMP_LO, F

        btfsc   STATUS, C
         incf   TEMP_HI, F

        movfw   TC77_HI
        addwf   TEMP_HI, F          ; result is in TEMP_HI:TEMP_LO

; divide results by 5

        clrf    TC77_HI
        clrf    TC77_LO
ML05
        movlw   D'5'                ; subtract 5 from TEMP_LO
        subwf   TEMP_LO, F

        btfsc   STATUS, C           ; was there a borrow?
         goto   ML10                ;  no, jump ahead
        movlw   D'1'
        subwf   TEMP_HI, F          ;  yes, borrow from TEMP_HI
        btfss   STATUS, C           ; was there a borrow from TEMP_HI?
         goto   ML15                ;  yes, we are done, jump ahead
ML10
        movlw   D'1'
```



```
        addwf   TC77_LO, F          ;  no, increment TC77_HI:TC77_LO
        btfsc   STATUS, C
         incf   TC77_HI, F

        goto    ML05                ; do it again

; add 32 (0x0200)

ML15
        movlw   0x02
        addwf   TC77_HI, F          ; result is in TC77_HI:TC77_LO

; end C to F conversion

; round result to integer value

ML20
        ; rotate right 3

        bcf     STATUS, C           ; clear carry bit
        rrf     TC77_HI, F          ; rotate right TC77_HI:TC77_LO 3 bits
        rrf     TC77_LO, F
        bcf     STATUS, C           ; clear carry bit
        rrf     TC77_HI, F
        rrf     TC77_LO, F
        bcf     STATUS, C           ; clear carry bit
        rrf     TC77_HI, F
        rrf     TC77_LO, F

        ; round

        movlw   D'1'
        addwf   TC77_LO, F
        btfsc   STATUS, C
         incf   TC77_HI, F

        ; rotate right 1

        bcf     STATUS, C           ; clear carry bit
        rrf     TC77_HI, F
        rrf     TC77_LO, F

;----------------------------------------

; convert into Binary Coded Decimal (BCD) format

        clrf    BCD_H               ; clear the BCD registers
        clrf    BCD_T
        clrf    BCD_O

        ; hundreds digit
ML25
        movlw   D'100'
        subwf   TC77_LO, W          ; subtract 100 (result goes into W)

        btfss   STATUS, C           ; was result negative?
         goto   ML30

        incf    BCD_H, F            ; no, increment BCD_H register
        movwf   TC77_LO             ; save result
        goto    ML25                 ; do it again
ML30
        movlw   D'10'               ; subtract 10 (result goes into W)
        subwf   TC77_LO, W
```

```
        btfss   STATUS, C           ; was result negative?
         goto   ML35

        incf    BCD_T, F            ; no, increment BCD_T register
        movwf   TC77_LO             ; save result
        goto    ML30                 ; do it again
ML35
        movfw   TC77_LO
        movwf   BCD_O               ; save result as BCD_O

; display on PICkit 1 FLASH Starter Kit LED's D7:D0

        movfw   BCD_O               ; move BCD Ones to TEMP
        movwf   TEMP

        swapf   BCD_T, W            ; swap BCD Tens nibbles
        iorwf   TEMP, W             ; inclusive or and store in TEMP

        movwf   LEDREG

        goto    MAINLOOP


;-----------------------------------------------------------------------
; Data EEPROM Memory (Section 8.0)
;
; PIC12F630/676 devices have 128 bytes of data EEPROM with address
; range 0x00 to 0x7F.

        ; Initialize Data EEPROM Memory locations

;        ORG     0x2100
;        DE      0x00, 0x01, 0x02, 0x03


;-----------------------------------------------------------------------
; Calibrating the Internal Oscillator (Section 9.2.5.1)
; Oscillator Calibration Register (OSCCAL) (Section 2.2.2.7)
;
; The below statements are placed here so that the program can be
; simulated with MPLAB(r) SIM or emulated with the ICD2 or ICE-2000.
;
; The programmer (PICkit(tm) or PRO MATE(r) II) will save the actual OSCCAL
; value in the device and restore it. The value below WILL NOT be
; programmed into the device.

        org     0x3ff
        retlw   0x80                ; Center Frequency


;-----------------------------------------------------------------------
        end                         ; end of program directive
;-----------------------------------------------------------------------
```

Note the following details of the code protection feature on Microchip devices:

- Microchip products meet the specification contained in their particular Microchip Data Sheet.
- Microchip believes that its family of products is one of the most secure families of its kind on the market today, when used in the intended manner and under normal conditions.
- There are dishonest and possibly illegal methods used to breach the code protection feature. All of these methods, to our knowledge, require using the Microchip products in a manner outside the operating specifications contained in Microchip's Data Sheets. Most likely, the person doing so is engaged in theft of intellectual property.
- Microchip is willing to work with the customer who is concerned about the integrity of their code.
- Neither Microchip nor any other semiconductor manufacturer can guarantee the security of their code. Code protection does not mean that we are guaranteeing the product as “unbreakable.”

Code protection is constantly evolving. We at Microchip are committed to continuously improving the code protection features of our products. Attempts to break Microchip’s code protection feature may be a violation of the Digital Millennium Copyright Act. If such acts allow unauthorized access to your software or other copyrighted work, you may have a right to sue for relief under that Act.

Trademarks

Printed on recycled paper.

QUALITY MANAGEMENT SYSTEM
CERTIFIED BY DNV
ISO/TS 16949:2002

Microchip received ISO/TS-16949:2002 quality system certification for its worldwide headquarters, design and wafer fabrication facilities in Chandler and Tempe, Arizona and Mountain View, California in October 2003. The Company’s quality system processes and procedures are for its PICmicro® 8-bit MCUs, KEELOQ® code hopping devices, Serial EEPROMs, microperipherals, nonvolatile memory and analog products. In addition, Microchip’s quality system for the design and manufacture of development systems is ISO 9001:2000 certified.

WORLDWIDE SALES AND SERVICE

AMERICAS

Corporate Office
2355 West Chandler Blvd.
Chandler, AZ 85224-6199
Tel: 480-792-7200
Fax: 480-792-7277
Technical Support: 480-792-7627
Web Address: http://www.microchip.com

Atlanta
3780 Mansell Road, Suite 130
Alpharetta, GA 30022
Tel: 770-640-0034
Fax: 770-640-0307

Boston
2 Lan Drive, Suite 120
Westford, MA 01886
Tel: 978-692-3848
Fax: 978-692-3821

Chicago
333 Pierce Road, Suite 180
Itasca, IL 60143
Tel: 630-285-0071
Fax: 630-285-0075

Dallas
4570 Westgrove Drive, Suite 160
Addison, TX 75001
Tel: 972-818-7423
Fax: 972-818-2924

Detroit
Tri-Atria Office Building
32255 Northwestern Highway, Suite 190
Farmington Hills, MI 48334
Tel: 248-538-2250
Fax: 248-538-2260

Kokomo
2767 S. Albright Road
Kokomo, IN 46902
Tel: 765-864-8360
Fax: 765-864-8387

Los Angeles
18201 Von Karman, Suite 1090
Irvine, CA 92612
Tel: 949-263-1888
Fax: 949-263-1338

San Jose
1300 Terra Bella Avenue
Mountain View, CA 94043
Tel: 650-215-1444
Fax: 650-961-0286

Toronto
6285 Northam Drive, Suite 108
Mississauga, Ontario L4V 1X5, Canada
Tel: 905-673-0699
Fax: 905-673-6509

ASIA/PACIFIC

Australia
Suite 22, 41 Rawson Street
Epping 2121, NSW
Australia
Tel: 61-2-9868-6733
Fax: 61-2-9868-6755

China - Beijing
Unit 706B
Wan Tai Bei Hai Bldg.
No. 6 Chaoyangmen Bei Str.
Beijing, 100027, China
Tel: 86-10-85282100
Fax: 86-10-85282104

China - Chengdu
Rm. 2401-2402, 24th Floor,
Ming Xing Financial Tower
No. 88 TIDU Street
Chengdu 610016, China
Tel: 86-28-86766200
Fax: 86-28-86766599

China - Fuzhou
Unit 28F, World Trade Plaza
No. 71 Wusi Road
Fuzhou 350001, China
Tel: 86-591-7503506
Fax: 86-591-7503521

China - Hong Kong SAR
Unit 901-6, Tower 2, Metroplaza
223 Hing Fong Road
Kwai Fong, N.T., Hong Kong
Tel: 852-2401-1200
Fax: 852-2401-3431

China - Shanghai
Room 701, Bldg. B
Far East International Plaza
No. 317 Xian Xia Road
Shanghai, 200051
Tel: 86-21-6275-5700
Fax: 86-21-6275-5060

China - Shenzhen
Rm. 1812, 18/F, Building A, United Plaza
No. 5022 Binhe Road, Futian District
Shenzhen 518033, China
Tel: 86-755-82901380
Fax: 86-755-8295-1393

China - Shunde
Room 401, Hongjian Building, No. 2
Fengxiangnan Road, Ronggui Town, Shunde District, Foshan City, Guangdong 528303, China
Tel: 86-757-28395507 Fax: 86-757-28395571

China - Qingdao
Rm. B505A, Fullhope Plaza,
No. 12 Hong Kong Central Rd.
Qingdao 266071, China
Tel: 86-532-5027355 Fax: 86-532-5027205

India
Divyasree Chambers
1 Floor, Wing A (A3/A4)
No. 11, O'Shaugnessey Road
Bangalore, 560 025, India
Tel: 91-80-22290061 Fax: 91-80-22290062

Japan
Benex S-1 6F
3-18-20, Shinyokohama
Kohoku-Ku, Yokohama-shi
Kanagawa, 222-0033, Japan
Tel: 81-45-471- 6166 Fax: 81-45-471-6122

Korea
168-1, Youngbo Bldg. 3 Floor
Samsung-Dong, Kangnam-Ku
Seoul, Korea 135-882
Tel: 82-2-554-7200 Fax: 82-2-558-5932 or 82-2-558-5934

Singapore
200 Middle Road
#07-02 Prime Centre
Singapore, 188980
Tel: 65-6334-8870 Fax: 65-6334-8850

Taiwan
Kaohsiung Branch
30F - 1 No. 8
Min Chuan 2nd Road
Kaohsiung 806, Taiwan
Tel: 886-7-536-4818
Fax: 886-7-536-4803

Taiwan
Taiwan Branch
11F-3, No. 207
Tung Hua North Road
Taipei, 105, Taiwan
Tel: 886-2-2717-7175 Fax: 886-2-2545-0139

EUROPE

Austria
Durisolstrasse 2
A-4600 Wels
Austria
Tel: 43-7242-2244-399
Fax: 43-7242-2244-393

Denmark
Regus Business Centre
Lautrup hoj 1-3
Ballerup DK-2750 Denmark
Tel: 45-4420-9895 Fax: 45-4420-9910

France
Parc d'Activite du Moulin de Massy
43 Rue du Saule Trapu
Batiment A - ler Etage
91300 Massy, France
Tel: 33-1-69-53-63-20
Fax: 33-1-69-30-90-79

Germany
Steinheilstrasse 10
D-85737 Ismaning, Germany
Tel: 49-89-627-144-0
Fax: 49-89-627-144-44

Italy
Via Quasimodo, 12
20025 Legnano (MI)
Milan, Italy
Tel: 39-0331-742611
Fax: 39-0331-466781

Netherlands
P. A. De Biesbosch 14
NL-5152 SC Drunen, Netherlands
Tel: 31-416-690399
Fax: 31-416-690340

United Kingdom
505 Eskdale Road
Winnersh Triangle
Wokingham
Berkshire, England RG41 5TU
Tel: 44-118-921-5869
Fax: 44-118-921-5820

02/17/04